Cross-Over Designs Not Variance Balanced

In the previous section variance balance crossover designs were presented where e all the treatment contrasts have the same precision and all carry-over contrasts have the same precision. These types of designs are used when the number of treatment periods is less than the number of treatments. These types of designs are used when the length of each treatment period is long and there are a large number of treatments that are being evaluated. The textbook lists three situations in medicine where this applies.

1) The chance that a patient will withdraw before completing the trial protocol increases the longer the overall length of the trial

2) In drug development, very long studies delay the drug registration. The clock starts ticking on how long a company has a patent on the drug from the time they first register the drug for testing, so a long development period can substantially reduce the time that they don't have to worry about competition from generics.

3) Ethics need to be considered in that trial participation shouldn't place too much burden on the patient.

If a variance balanced crossover design is not feasible, then designs that approximate variance balance should be used. Davis and Hall in 1969 presented cyclic designs and Patterson and Lucas (1962), presented a number of partially balanced incomplete block designs that are approximate variance balanced. In approximate variance balanced incomplete block designs, no treatment appears more than once in a block and an ordered treatment pair occurs with the same frequency as its reversed-ordered pair (AB vs. BA). For example, if there are 6 treatments (A-F) and 4 periods, the following design would fulfill the above requirements.

	Block 1				Block 2				Block3			
Subject->	1	2	3	4	1	2	3	4	1	2	3	4
Period 1	A	D	B	E	B	E	C	F	C	F	A	D
Period 2	D	E	A	B	E	F	B	C	F	D	C	A
Period 3	B	A	E	D	C	B	F	E	A	C	D	F
Period 4	E	B	D	A	F	C	E	B	D	A	F	C

Note that Treatments C and F don't appear in Block 1, Treatments A and D don't appear in Block 2, and Treatments B and E don't appear in Block 3. Also note that within each block, each ordered treatment pair occurs with the same frequency. Those pairs of treatments that occur the most number of times (n1) are called first order associates and those that occur fewer times (n2) are called second order associates. For example, AD would be a first order associate, because it appears 2 times, while DE would be considered a second order associate, because it appears only once. The first order associates in this design are AD, FC, and EB. Also note that the number of first order associates is the same for Block 1 (AD, EB), Block 2 (FC, EB) and Block 3 (AD, FC). We will see later what implication this has on estimates and standard errors.

Three-treatment, two-period

The first design that we'll look at in depth is Koch's design, which is a three-treatment, two-period design. With the two-treatment, two-period crossover design that we looked at previously, carryover effect was completely aliased with sequence and treatment*period, because we only had four degrees of freedom for overall mean, treatment, period and sequence. With Koch's design, we now have 12 degrees of freedom available.

	Block 1		Block 2		Block 3	
Subject ->	1	2	1	2	1	2
Period 1	A	B	A	C	B	C
Period 2	B	A	C	A	C	B

With this type of design, it is more efficient treating patient as random rather than fixed. The textbook presents the results from Mead (1988) of a three-treatment two-period crossover trial. They looked at the standard errors of the treatment and carryover effects when patients were treated as fixed and then as random. The standard errors in the random patient analysis were substantially smaller than in the fixed analysis.

Analysis of 6 Treatment, 4 Period Design

We'll now run through an example from the "Cross-Over Experiments" book using the design presented earlier. The study involved 12 cows, so there was only one cow per sequence.

The linear model for this experiment

$$y_{ijk} = \mu + B_i + \xi_{i(k)} + \pi_{i(j)} + \tau_{t(ij)} + \delta_j \lambda_{r(ij)} + e_{ijk}$$

μ = mean
B_i = block effect for block i
$\xi_{i(k)}$ = random subject effect nested within block
$\pi_{i(j)}$ = period effect for period j
$\tau_{t(ij)}$= direct treatment effect for treatment t
$\lambda_{r(ij)}$ = the first-order "carryover" effect due to treatment r, where r is the treatment applied the previous period
δ_j = an indicator variable: 0 if it is the first period
1 if the period is greater than 1

The following SAS program was run. Note that we are using the delta indicator variable that we saw earlier and that we still need to code the period 1 carryover arbitrarily to one of the other treatments. It doesn't matter which one, because it is multiplied by the 0 delta. Because cow is the same as sequence, we do not need to include sequence in the model. Because delta is not a fixed effect, we cannot use the lsmeans statement in order to look at carryover differences. We need to set up estimate statements for the different carryover comparisons.

```
data fcm;
input block COW period treat $ carry $ fcm @@;
delta=(carry^='0');
if carry = '0' then carry='F';
cards;

1 11 1 A 0 38.7 1 11 2 D A 37.4 1 11 3 B D 34.3 1 11 4 E B 31.3
1 12 1 B 0 48.9 1 12 2 A B 46.9 1 12 3 E A 42.0 1 12 4 D E 39.6
1 13 1 E 0 34.6 1 13 2 B E 32.3 1 13 3 D B 28.5 1 13 4 A D 27.1
1 14 1 D 0 35.2 1 14 2 E D 33.5 1 14 3 A E 28.4 1 14 4 B A 25.1
2 21 1 D 0 32.9 2 21 2 A D 33.1 2 21 3 F A 27.5 2 21 4 C F 25.1
2 22 1 F 0 30.4 2 22 2 D F 29.5 2 22 3 C D 26.7 2 22 4 A C 23.1
2 23 1 C 0 30.8 2 23 2 F C 29.3 2 23 3 A F 26.4 2 23 4 D A 23.2
2 24 1 A 0 25.7 2 24 2 C A 26.1 2 24 3 D C 23.4 2 24 4 F D 18.7
3 31 1 E 0 25.4 3 31 2 F E 26.0 3 31 3 B F 23.9 3 31 4 C B 19.9
3 32 1 B 0 21.8 3 32 2 E B 23.9 3 32 3 C E 21.7 3 32 4 F C 17.6
3 33 1 F 0 21.4 3 33 2 C F 22.0 3 33 3 E C 19.4 3 33 4 B E 16.6
3 34 1 C 0 22.8 3 34 2 B C 21.0 3 34 3 F B 18.6 3 34 4 E F 16.1
;
RUN;

options ps=70 ls=70;

PROC MIXED;
CLASS block cow period carry treat;;
MODEL fcm=block period(block) treat carry*delta/solution ;
RANDOM cow(block);

estimate 'carry A-B' carry*delta 1 -1 0 0 0 0;
estimate 'carry A-C' carry*delta 1 0 -1 0 0 0;
estimate 'carry A-D' carry*delta 1 0 0 -1 0 0;
estimate 'carry A-E' carry*delta 1 0 0 0 -1 0;
estimate 'carry A-F' carry*delta 1 0 0 0 0 -1;
estimate 'carry B-C' carry*delta 0 1 -1 0 0 0;
estimate 'carry B-D' carry*delta 0 1 0 -1 0 0;
estimate 'carry B-E' carry*delta 0 1 0 0 -1 0;
estimate 'carry B-F' carry*delta 0 1 0 0 0 -1;
estimate 'carry C-D' carry*delta 0 0 1 -1 0 0;
estimate 'carry C-E' carry*delta 0 0 1 0 -1 0;
estimate 'carry C-F' carry*delta 0 0 1 0 0 -1;
estimate 'carry D-E' carry*delta 0 0 0 1 -1 0;
estimate 'carry D-F' carry*delta 0 0 0 1 0 -1;
estimate 'carry E-F' carry*delta 0 0 0 0 1 -1;

lsmeans treat/pdiff;

run;
```

Following is selected output from the analysis.

Solution for Fixed Effects

Effect	carry	treat	block	period	Estimate	Standard Error	DF
Intercept					17.0703	2.1932	9
block			1		13.0002	3.0497	9
block			2		4.7727	3.0497	9
block			3		0	.	.
period(block)			1	1	8.8186	0.6622	17
period(block)			1	2	6.7500	0.5236	17
period(block)			1	3	2.5250	0.5236	17
period(block)			1	4	0	.	.
period(block)			2	1	7.6356	0.6045	17
period(block)			2	2	6.9750	0.5236	17
period(block)			2	3	3.4750	0.5236	17
period(block)			2	4	0	.	.
period(block)			3	1	5.4188	0.6045	17
period(block)			3	2	5.6750	0.5236	17
period(block)			3	3	3.3500	0.5236	17
period(block)			3	4	0	.	.
treat		A			0.4078	0.4194	17
treat		B			0.1895	0.4194	17
treat		C			0.7429	0.3883	17
treat		D			0.7349	0.4194	17
treat		E			0.5114	0.4194	17
treat		F			0	.	.

Solution for Fixed Effects

Effect	carry	treat	block	period	t Value	Pr > |t|
Intercept					7.78	<.0001
block			1		4.26	0.0021
block			2		1.56	0.1520
block			3		.	.
period(block)			1	1	13.32	<.0001
period(block)			1	2	12.89	<.0001
period(block)			1	3	4.82	0.0002
period(block)			1	4	.	.
period(block)			2	1	12.63	<.0001
period(block)			2	2	13.32	<.0001
period(block)			2	3	6.64	<.0001
period(block)			2	4	.	.
period(block)			3	1	8.96	<.0001
period(block)			3	2	10.84	<.0001
period(block)			3	3	6.40	<.0001
period(block)			3	4	.	.
treat		A			0.97	0.3445
treat		B			0.45	0.6571
treat		C			1.91	0.0727
treat		D			1.75	0.0977
treat		E			1.22	0.2393
treat		F			.	.

Solution for Fixed Effects

Effect	carry	treat	block	period	Estimate	Standard Error	DF
delta*carry	A				-0.2437	0.5056	17
delta*carry	B				0.08284	0.5056	17
delta*carry	C				0.1716	0.4681	17
delta*carry	D				0.9146	0.5056	17
delta*carry	E				0.2207	0.5056	17

```
delta*carry    F                                  0          .     .
```

Solution for Fixed Effects

Effect	carry	treat	block	period	t Value	Pr > |t|
delta*carry	A				-0.48	0.6360
delta*carry	B				0.16	0.8718
delta*carry	C				0.37	0.7184
delta*carry	D				1.81	0.0882
delta*carry	E				0.44	0.6680
delta*carry	F				.	.

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
block	2	9	10.78	0.0041
period(block)	8	17	61.67	<.0001
treat	5	17	1.13	0.3821
delta*carry	5	17	1.36	0.2876

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
carry A-B	-0.3265	0.5056	17	-0.65	0.5271
carry A-C	-0.4153	0.5056	17	-0.82	0.4228
carry A-D	-1.1583	**0.4681**	17	-2.47	0.0242
carry A-E	-0.4643	0.5056	17	-0.92	0.3713
carry A-F	-0.2437	0.5056	17	-0.48	0.6360
carry B-C	-0.08876	0.5056	17	-0.18	0.8627
carry B-D	-0.8317	0.5056	17	-1.65	0.1183
carry B-E	-0.1378	**0.4681**	17	-0.29	0.7720
carry B-F	0.08284	0.5056	17	0.16	0.8718
carry C-D	-0.7430	0.5056	17	-1.47	0.1600
carry C-E	-0.04906	0.5056	17	-0.10	0.9238
carry C-F	0.1716	**0.4681**	17	0.37	0.7184
carry D-E	0.6939	0.5056	17	1.37	0.1878
carry D-F	0.9146	0.5056	17	1.81	0.0882
carry E-F	0.2207	0.5056	17	0.44	0.6680

Least Squares Means

Effect	treat	Estimate	Standard Error	DF	t Value	Pr > |t|
treat	A	27.7642	1.2545	17	22.13	<.0001
treat	B	27.5459	1.2545	17	21.96	<.0001

The Mixed Procedure

Least Squares Means

Effect	treat	Estimate	Standard Error	DF	t Value	Pr > |t|
treat	C	28.0993	1.2545	17	22.40	<.0001
treat	D	28.0913	1.2545	17	22.39	<.0001
treat	E	27.8678	1.2545	17	22.21	<.0001
treat	F	27.3564	1.2545	17	21.81	<.0001

Differences of Least Squares Means

Effect	treat	_treat	Estimate	Standard Error	DF	t Value	Pr > |t|

```
treat   A    B       0.2184    0.4194    17     0.52    0.6093
treat   A    C      -0.3351    0.4194    17    -0.80    0.4353
treat   A    D      -0.3271    0.3883    17    -0.84    0.4113
treat   A    E      -0.1036    0.4194    17    -0.25    0.8079
treat   A    F       0.4078    0.4194    17     0.97    0.3445
treat   B    C      -0.5534    0.4194    17    -1.32    0.2044
treat   B    D      -0.5454    0.4194    17    -1.30    0.2108
treat   B    E      -0.3220    0.3883    17    -0.83    0.4185
treat   B    F       0.1895    0.4194    17     0.45    0.6571
treat   C    D     0.008004    0.4194    17     0.02    0.9850
treat   C    E       0.2315    0.4194    17     0.55    0.5882
treat   C    F       0.7429    0.3883    17     1.91    0.0727
treat   D    E       0.2235    0.4194    17     0.53    0.6010
treat   D    F       0.7349    0.4194    17     1.75    0.0977
treat   E    F       0.5114    0.4194    17     1.22    0.2393
```

Some things to note about this analysis. SAS uses the set-to-zero convention, setting the last level of the effect, so that all other levels are contrasted against that last effect. Note that for the estimates for both the treat and delta*carry effects, the standard error estimate for treatment C is smaller than for the other treatments, indicating that this is a more precise estimate. If we go back and look at the study design, we see that FC is the only first order association involving treatment F, which means that this difference occurs more often, explaining the lower variance. When we look at the estimates for the carryover differences, we see that the standard errors for the AD, BE, and CF are all smaller than for the others. These three pairs of treatments are the first order association pairs in this design, meaning they occur more often than the other pairs and therefore have smaller standard errors. The standard errors are the same for the first order association pairs, because there are an equal number in each of the three blocks.

Cyclic Designs

Cyclic crossover designs are extended cyclic incomplete block designs and are called the Designs of Davis and Hall, who proposed this class of designs in 1969. They are called cyclic because given the sequence of treatments in the different time periods for the first subject of each block, the sequence for the second subject is obtained by going to the next letter of the alphabet, except when the letter corresponding to the last treatment is reached, you go back to the first letter. The block size is always equal to the number of treatments. Note that the cyclic design does not have the requirement that the approximate variance balanced incomplete block design has that an ordered treatment pair occurs with the same frequency as its reversed-ordered pair (AB vs. BA). An example 6-treatment, 3-period, 2-block design is given below, with starting sequences for the two blocks of ADE and AFB.

Block 1						
Subject->	1	2	3	4	5	6
Period 1	A	B	C	D	E	F
Period 2	D	E	F	A	B	C
Period 3	E	F	A	B	C	D
Block 2						
Subject->	1	2	3	4	5	6
Period 1	A	B	C	D	E	F
Period 2	F	A	B	C	D	E

Period 3	B	C	D	E	F	A

With this design, there are nine treatment pairs that have a reversed-ordered pair. Of these nine treatment pairs, three have both their pair and reversed-ordered pair appearing in the same block (AD, BE, CF). The other six (AB, AF, BC, CD, DE, EF) have their reversed-ordered pair appearing in a different block from their pair. The remaining six pairs (AC, AE, BD, BF, CE, DF) do not have a reversed-ordered pair.

We'll run through an example of this design with the following SAS program.

```
data davhall;
input block subj period treat $ carry $ resp @@;
delta=(carry^='0');
if carry='0' then carry='F';
cards;
1 11 1 A 0 38.7 1 11 2 D A 37.4 1 11 3 E D 34.3
1 12 1 B 0 48.9 1 12 2 E B 46.9 1 12 3 F E 42.0
1 13 1 C 0 35.2 1 13 2 F C 33.5 1 13 3 A F 28.4
1 14 1 D 0 34.6 1 14 2 A D 32.3 1 14 3 B A 28.5
1 15 1 E 0 32.9 1 15 2 B E 33.1 1 15 3 C B 27.5
1 16 1 F 0 30.4 1 16 2 C F 29.5 1 16 3 D C 26.7
2 21 1 A 0 25.7 2 21 2 F A 26.1 2 21 3 B F 23.4
2 22 1 B 0 30.8 2 22 2 A B 29.3 2 22 3 C A 26.4
2 23 1 C 0 25.4 2 23 2 B C 26.0 2 23 3 D B 23.9
2 24 1 D 0 21.8 2 24 2 C D 23.9 2 24 3 E C 21.7
2 25 1 E 0 21.4 2 25 2 D E 22.0 2 25 3 F D 19.4
2 26 1 F 0 22.8 2 26 2 E F 21.0 2 26 3 A E 18.6
;

PROC MIXED;
CLASS block SUBJ period carry treat;;
MODEL resp=block period(block) treat carry*delta/solution ;
RANDOM subj(block);
estimate 'carry A-B' carry*delta 1 -1 0 0 0 0;
estimate 'carry A-C' carry*delta 1 0 -1 0 0 0;
estimate 'carry A-D' carry*delta 1 0 0 -1 0 0;
estimate 'carry A-E' carry*delta 1 0 0 0 -1 0;
estimate 'carry A-F' carry*delta 1 0 0 0 0 -1;
estimate 'carry B-C' carry*delta 0 1 -1 0 0 0;
estimate 'carry B-D' carry*delta 0 1 0 -1 0 0;
estimate 'carry B-E' carry*delta 0 1 0 0 -1 0;
estimate 'carry B-F' carry*delta 0 1 0 0 0 -1;
estimate 'carry C-D' carry*delta 0 0 1 -1 0 0;
estimate 'carry C-E' carry*delta 0 0 1 0 -1 0;
estimate 'carry C-F' carry*delta 0 0 1 0 0 -1;
estimate 'carry D-E' carry*delta 0 0 0 1 -1 0;
estimate 'carry D-F' carry*delta 0 0 0 1 0 -1;
estimate 'carry E-F' carry*delta 0 0 0 0 1 -1;
lsmeans treat/pdiff;
run;
```

Note that the model, except for some changes in variable names, is the same as the previous example. Following is selected output.

Solution for Fixed Effects

Effect	carry	treat	block	period	Estimate	Standard Error	DF
Intercept					20.9911	2.1307	10
block			1		9.0000	2.8063	10
block			2		0	.	.
period(block)			1	1	6.5227	0.7014	10
period(block)			1	2	4.2167	0.4700	10
period(block)			1	3	0	.	.
period(block)			2	1	3.3894	0.7014	10
period(block)			2	2	2.4833	0.4700	10
period(block)			2	3	0	.	.
treat		A			-0.5407	0.6096	10
treat		B			1.0802	0.6270	10
treat		C			0.3406	**0.5903**	10
treat		D			0.4062	0.6270	10
treat		E			0.3307	0.6096	10
treat		F			0	.	.
delta*carry	A				0.3288	0.8179	10
delta*carry	B				1.4290	0.8434	10
delta*carry	C				1.3963	**0.7017**	10
delta*carry	D				1.5530	0.8434	10
delta*carry	E				1.1291	0.8179	10
delta*carry	F				0	.	.

Solution for Fixed Effects

Effect	carry	treat	block	period	t Value	Pr > |t|
Intercept					9.85	<.0001
block			1		3.21	0.0094
block			2		.	.
period(block)			1	1	9.30	<.0001
period(block)			1	2	8.97	<.0001
period(block)			1	3	.	.
period(block)			2	1	4.83	0.0007
period(block)			2	2	5.28	0.0004
period(block)			2	3	.	.
treat		A			-0.89	0.3960
treat		B			1.72	0.1157
treat		C			0.58	0.5767
treat		D			0.65	0.5317
treat		E			0.54	0.5994
treat		F			.	.
delta*carry	A				0.40	0.6962
delta*carry	B				1.69	0.1210
delta*carry	C				1.99	0.0746
delta*carry	D				1.84	0.0954
delta*carry	E				1.38	0.1975
delta*carry	F				.	.

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
block	1	10	14.60	0.0034
period(block)	3	10	41.30	<.0001
treat	5	10	1.52	0.2668
delta*carry	5	10	1.48	0.2798

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
carry A-B	-1.1003	0.8179	10	-1.35	0.2083
carry A-C	-1.0676	0.8434	10	-1.27	0.2343
carry A-D	-1.2243	**0.7017**	10	-1.74	0.1116
carry A-E	-0.8003	0.8434	10	-0.95	0.3650
carry A-F	0.3288	0.8179	10	0.40	0.6962
carry B-C	0.03270	0.8179	10	0.04	0.9689
carry B-D	-0.1240	0.8434	10	-0.15	0.8860
carry B-E	0.2999	**0.7017**	10	0.43	0.6781
carry B-F	1.4290	0.8434	10	1.69	0.1210
carry C-D	-0.1567	0.8179	10	-0.19	0.8519
carry C-E	0.2672	0.8434	10	0.32	0.7579
carry C-F	1.3963	**0.7017**	10	1.99	0.0746
carry D-E	0.4239	0.8179	10	0.52	0.6155
carry D-F	1.5530	0.8434	10	1.84	0.0954
carry E-F	1.1291	0.8179	10	1.38	0.1975

Least Squares Means

Effect	treat	Estimate	Standard Error	DF	t Value	Pr > |t|
treat	A	28.3676	1.4452	10	19.63	<.0001
treat	B	29.9884	1.4452	10	20.75	<.0001
treat	C	29.2489	1.4452	10	20.24	<.0001
treat	D	29.3145	1.4452	10	20.28	<.0001
treat	E	29.2389	1.4452	10	20.23	<.0001
treat	F	28.9083	1.4452	10	20.00	<.0001

Differences of Least Squares Means

Effect	treat	_treat	Estimate	Standard Error	DF	t Value	Pr > |t|
treat	A	B	-1.6208	0.6096	10	-2.66	0.0240
treat	A	C	-0.8813	0.6270	10	-1.41	0.1902
treat	A	D	-0.9469	**0.5903**	10	-1.60	0.1398
treat	A	E	-0.8713	0.6270	10	-1.39	0.1948
treat	A	F	-0.5407	0.6096	10	-0.89	0.3960
treat	B	C	0.7395	0.6096	10	1.21	0.2530
treat	B	D	0.6739	0.6270	10	1.07	0.3077
treat	B	E	0.7495	**0.5903**	10	1.27	0.2330
treat	B	F	1.0802	0.6270	10	1.72	0.1157
treat	C	D	-0.06560	0.6096	10	-0.11	0.9164
treat	C	E	0.009961	0.6270	10	0.02	0.9876
treat	C	F	0.3406	**0.5903**	10	0.58	0.5767
treat	D	E	0.07556	0.6096	10	0.12	0.9038
treat	D	F	0.4062	0.6270	10	0.65	0.5317
treat	E	F	0.3307	0.6096	10	0.54	0.5994

Note that with this design, that there is even more variability in the standard errors. Looking at the treatment and carryover effects, we see that Treatment C has the smallest standard error. Again, this is because SAS uses the set-to-zero convention, setting the last level of the effect to zero, so that all other levels are contrasted against that last effect (Block 2 and Treatment F). Because both the CF and FC pairs appear in Block 1, Treatment C has the most precise estimate. AF and EF are the only other pairs that also have a reversed-ordered pair. However, their pair and reversed-ordered pair appear in different blocks, making their estimates less precise. Finally, the least precise estimates are for those treatments that are paired only once with F (BF and DF). Looking at the treatment comparisons, the AD, BE and CF estimates are the most precise. This is because all three of these comparisons have both the pair and reversed-ordered pair appearing only in Block 1. The comparisons that have their pair and reversed-order pairs appearing in different blocks have less precision (AB, AF, BC, CD, DE, EF) and the pairs that do not have a reverse-ordered pair are the least precise (AC, AE, BD, BF, CE, DF).